

Summary count by month with COUNTIFS

| Issue | Date | Priority | Month | Count | Priority | 1 | 2 | 3 |
|-----------------|-----------|----------|-----------|-------|----------|---|---|---|
| Hardware issue | 2/1/2015 | 1 | 1/1/2015 | 8 | 1 | 6 | 2 | 0 |
| Software issue | 5/1/2015 | 3 | 2/1/2015 | 8 | 2 | 4 | 2 | 2 |
| Hardware issue | 11/1/2015 | 2 | 3/1/2015 | 13 | 3 | 7 | 2 | 4 |
| Unknown | 7/1/2015 | 1 | 4/1/2015 | 8 | 4 | 3 | 3 | 2 |
| User error | 2/1/2015 | 3 | 5/1/2015 | 9 | 5 | 4 | 2 | 3 |
| User error | 1/1/2015 | 1 | 6/1/2015 | 6 | 6 | 2 | 1 | 3 |
| Network problem | 3/1/2015 | 3 | 7/1/2015 | 8 | 7 | 1 | 3 | 4 |
| User error | 8/1/2015 | 3 | 8/1/2015 | 7 | 8 | 3 | 2 | 2 |
| User error | 3/1/2015 | 1 | 9/1/2015 | 13 | 9 | 5 | 4 | 4 |
| Software issue | 3/1/2015 | 1 | 10/1/2015 | 4 | 10 | 0 | 2 | 2 |
| Network problem | 1/1/2015 | 1 | 11/1/2015 | 11 | 11 | 5 | 1 | 5 |
| Hardware issue | 9/1/2015 | 3 | 12/1/2015 | 5 | 12 | 2 | 2 | 1 |

Generic formula

```
=COUNTIFS(dates, ">=" & A1, dates, "<" & EDATE(A1, 1))
```

Summary

To create a summary count by month, you can use the COUNTIFS function and the EDATE function with two criteria. In the example shown, the formula in G5 is:

```
=COUNTIFS(dates, ">=" & F5, dates, "<" & EDATE(F5, 1))
```

Explanation

In this example, we have a list of 100 issues in Columns B to D. Each issue has a date and priority. We are also using the [named range](#) "dates" for C5:C104 and "priorities" for D5:D105. Starting in column F, we have a summary table that shows a total count per month, followed by a total count per month per priority.

We are using the [COUNTIFS function](#) to generate a count. The first column of the summary table (F) is a date for the first of each month in 2015. To generate a total count per month, we need to supply criteria that will isolate all the issues that appear in each month.

Since we have actual dates in column F, we can construct the criteria we need using the date itself, and a second date created with the [EDATE function](#). These two criteria appear inside COUNTIFS like so:

```
dates, ">=" & F5, dates, "<" & EDATE(F5, 1)
```

Roughly translated: "dates greater than or equal to the date in F5 and less than the date in F5 plus one month". This is a convenient way to generate "brackets" for each month based on a single date.

When the formula is copied down column G, COUNTIFS generates the correct count for each month.

Note: if you don't want to see full dates in column F, just apply the [custom date formats](#) "mmm" or "mmmm" to display the month names only.

With Priority

To generate a count by priority, we need to extend criteria. The formula in H5 is:

```
=COUNTIFS(dates, ">=" & $F5, dates, "<" & EDATE($F5, 1), priorities, H$4)
```